
1. Introduction 

In this paper we present a first approach to VAL, an implementation-independent agent-
oriented programming language. It allows developers to write agent representation code in a 
conceptual level, focusing specifically upon the agent’s definition requirements. VAL code is 
not in any way related to the agent system’s architecture and implementation, allowing agent 
definitions to be portable between systems. 

VAL can be used in cooperation with agent systems that maintain an internal symbolic 
representation for the virtual world and the agent’s internal state. By the term symbolic 
representation we refer to an unordered set of statements that denote the agent’s and the 
world’s valid properties at any time. Statements can be, for instance, predicate logic Prolog-
style terms. VAL agent definitions are based entirely on these properties and do not descend 
to the implementation level, allowing for a high-level agent-oriented development process. 
This is one of VAL’s most notable characteristics. 

In the following section we will discuss upon several key topics including agent systems, 
agent languages, virtual reality and how research and combination of the above lead to the 
need for an agent-oriented programming language. In section 3 we will attempt an 
introduction to VAL through simple examples. A formal presentation of the syntax and 
semantics of VAL will be given in section 4, VAL based systems architecture in section 5, 
while in section 6 the reader can follow a step-by-step illustrative example of developing a 
complete agent using VAL and it’s tools. The paper concludes with implementation details 
and future plans. 

2. Towards a Virtual reality Agent oriented Language 

2.1 Agent Systems and Agent Languages 

Today there is a variety of architectures for agents and agent systems [1, 2]. In this paper, 
when refering to agent systems we will use the following abstraction (please see figure 2.1): 
Agent systems are viewed as a set of cooperating software components. One of the 
components is the virtual world implementation and will be referred to as the Virtual World 
Component (VWC), while the others are the agent implementations and will be referred to as 
the Intelligent Agent Components (IAC). 



Figure 2.1 – The abstraction used for Agent Systems 

The Agent Components are usually predefined and cannot change without significant 
programming work. Some behaviour adjustment and configuration might be available, being, 
however, far from providing a means to 'reprogram' an agent, make it capable of performing a 
totally different task within the system. 

The reason is that in most cases, agent systems are built to be used in predetermined domains 
with specific problem solving needs. Each implementation is created manually using standard 
programming languages and techniques. 

A remedy to the above situation is languages for creating agents [3, 4, 5]. Research during the 
past decade, beginning with Shoham’s Agent0 language, [6], has resulted in a number of 
agent languages which combined with tools for producing implementation specific agent 
representation data can aid the developer in designing agents in a more abstract and less 
architecture bound way. 

2.2 Virtual Reality and Artificial Intelligence 

Another key issue in agent systems is virtual environment representation [7, 8]. Research in 
the field of Virtual Reality has resulted in a variety of architectures and tools for creating 
virtual worlds. Virtual Reality Modeling Language (VRML) is an example of such a tool, 
providing developers with the means to create virtual worlds and to access them via networks 
easily, using simple applications such as WWW browsers [9]. 

During the past few years, the need for combining Artificial Intelligence with Virtual Reality 
has become obvious [10, 11]. Virtual Reality systems without an underlying intelligent 
control mechanism may be impressive but are short of any significant research potential, 
while on the other hand, AI applications and systems without an illustrative visualization 
mechanism are hard to approach, comprehend and aid in research. 

Recent research in University of Piraeus has led to a framework integrating VRML as a 
virtual environment representation provider with Artificial Intelligence implemented through 
an intelligent agent system [10]. 

2.3 Motivation and specifics of VAL 

In the first demonstrative implementation based on that architecture, the agent's abilities were 
defined inside a Prolog source code module. The code was structured according to the 
architecture's needs and was, in fact, quite complicated. When the need for real world agent 
applications rose, it became clear that apart from designing new agents in a conceptual layer, 
the developer had to deal with the implementation details for the specific framework. This 
was at least difficult and time consuming. Furthermore, no actual research could be carried 
out due to the lack of an efficient agent design and testing mechanism. 

VAL is a means to avoid the complications mentioned above. VAL was initially developed as 
a tool to define agent’s abilities for [10], but can be extended towards a complete agent 
language. VAL based systems provide tools for producing implementation specific agent 
representation data from VAL code. In such systems, the agent developer is never concerned 
with the underlying architecture, just as, for instance, Java developers need not be aware of 
highly technical hardware specifics. 

IAC VWC 



3. Presentation of VAL through simple examples 

In this section, the usage of VAL is depicted through several simple examples. VAL currently 
supports definition of an agent’s abilities only. However, as it will become obvious in the 
next section, it can easily be extended to provide the capability of defining other aspects of an 
agent such as its goals, its reasoning abilities, etc. 

The following examples are based on the assumption that the Virtual World and the Agent 
are completely represented by the properties seen on tables 3.1 and 3.2, with the appropriate 
meanings. Furthermore, we assume that the agent’s meta-interpreter (or other type of 
reasoning engine) can perform the append action on list-type terms. 

 

at(position) The agent is located on position in the Virtual World 

connects(position1, position2) position1 is connected to position2 in the Virtual World 

is_at(item, position) item is located on position 

Table 3.1 – The Virtual World representation properties 

holding(items) The list of things the agent is holding is items 

Table 3.2 – The Agent internal state properties 

Example 1 – A simple ability 

agent johnie { 
 abilities { 
  go(NewPos) { 
   preconditions { 

   at(CurrentPos), 
    connects(NewPos, CurrentPos) 

}   
   effects { 
    - at(CurrentPos), 
    + at(NewPos) 

} 
  } 

} 
} 
 
This example shows an agent with just one ability, the ‘go’ ability. As the example 
demonstrates, VAL programs are structured in blocks. The largest one is the agent block, 
which states the agent’s name and contains all the remaining VAL code. An agent block 
completely defines an agent. A second level block, the abilities block, completely defines the 
agent’s abilities. The abilities block contains several ability definition blocks beginning with 
an ability name. Each ability definition block is divided into two sections, the preconditions 
section and the effects section.  

The code inside the preconditions section states that the ‘go’ action can only be applied when 
the two properties at(CurrentPos) and connects(NewPos, CurrentPos) are true in the 



environment’s description. The effects section in the ‘go’ ability definition block of the above 
program contains two statements beginning with a ‘-’ and a ‘+’ sign respectively. Statements 
beginning with a minus sign declare properties that are to be retracted from the virtual 
environment’s description, while those beginning with a plus sign declare properties to be 
asserted. When the ‘go’ action is applied, the at(CurrentPos) property is retracted from the 
virtual environment’s description and the at(NewPos) property is asserted, meaning that after 
the action is completed, the agent will have ceased being located on CurrentPos and will 
have been transferred on NewPos which is defined as an argument to the action. 

 
Example 2 – A more complex ability 

agent picker { 
 abilities { 
  pickup(Item) { 
   preconditions { 
    at(CurrentPos), 

 is_at(Item, CurrentPos) 
} 

   effects { 
    holding(Items), 
    append(Items, Item, NewItems), 
    - is_at(Item, CurrentPos), 
    \- holding(Items), 
    \+ holding(NewItems) 

} 
  } 

} 
} 
 

The notation introduced in this example is statements that follow a ‘\+’ or a ‘\-‘ sign and 
those that are not preceded by any sign. The former are statements that are to be asserted or 
retracted respectively from the agent’s internal state and not from the world description, and 
the latter are statements to be solved by the agent’s meta-interpreter, that is, either variables 
to be instantiated (the ‘holding(Items)’ statement) or instructions to be executed.(the 
‘append(Items, Item, NewItems)’ statement). The reason for the ‘\+’ and ‘\-‘ operators is that 
the agent’s reasoning can result in states that cannot be observed in the world, therefore their 
properties need to be explicitly added or removed to and from its internal state. 

The examples so far have provided the reader with a first glance at VAL, intuitively depicting 
VAL programs structure, how to define agent abilities through effects and preconditions as 
well as some special notation. A detailed analysis of VAL’s syntax and semantics follows in 
the next section. 

4. Syntax and Semantics of VAL 

The structure of VAL programs is based on the C/C++ language blocks notation. 
Furthermore, the world and agent representation statements are Prolog-style predicate logic 
symbols. A formal definition of VAL’s grammar follows in figure 4.1. 

 
agent_block:  ‘{‘ abilities_block ‘}’ 



abilities_block: ‘abilities’ ‘{‘ abilities_list ‘}’ 
abilities_list:  ability { ability } 
ability:   ability_name ‘{‘ preconditions_block effects_block ‘}’ 
ability_name:  functor 
preconditions_block: ‘preconditions’ ‘{‘ preconditions_list ‘}’ 
effects_block:  ‘effects’ ‘{‘ effects_list ‘}’ 
preconditions_list: [ { precondition ‘,’ } precondition ] 
effects_list:  [ { effect ‘,’ } effect ] 
precondition:  statement 
effect:   statement | ‘+’ functor | ‘-‘ functor | ‘\+‘ functor | ‘\-‘ functor 

Figure 4.1 – VAL syntax (EBNF notation) 

Again, functors are valid Prolog functors and statements are valid Prolog terms, that is, 
identifiers, variables, numbers, functors, lists and operator expressions. 

As far as VAL’s semantics are concerned, the following rules apply. First, each statement 
appearing inside a preconditions block needs to be true in order for an ability to be applicable 
at a certain point in time. Second, statements appearing inside an effects block are treated 
according to the operator that precedes them. For a definition of the operator’s meanings 
please refer to table 4.2 below. 

 

Operator Meaning 

No operator Statement is solved 

+ Property is asserted to the world’s description 

- Property is retracted from the world’s description 

\+ Property is asserted to the agent’s internal state 

\- Property is retracted from the agent’s internal state 

Table 4.2 – VAL effect-specific operators 

5. VAL based systems architecture  

Agent systems that support cooperation with VAL defined agents are based on the following 
architecture. 

The Upper Module (Logical Decision, Abstract Definition) takes all the decisions at an 
abstract logical level and therefore defines abstract logical actions, e.g. go to next room, 
pickup an object, etc. The Intermediate Module (Domain specific Instantiation, Detailed 
Definition) computes the values of certain parameters needed for an action to be actually 
performed, e.g. to move from one room to another we must compute the corresponding 
coordinates and maybe, if needed, an intermediate path. The Lower Module (Action and 
Object Visualisation/Execution) takes care of the actual manipulation of an object, i.e. the 
execution of the action. 

The Upper Module is defined using VAL code. Using this definition and a VAL compiler, an 
Intermediate Module is produced. This module is the actual, agent system oriented 
implementation of the agent oriented attributes and processes defined in the Upper Module. 



The Lower Module, which is the more implementation bound from all three, is the 
visualisation of the actions, that is, the ‘realisation’ of the abstract, conceptually defined 
agent, its form, abilities, decisions, actions and plans. 

Figure 5.1 – VAL based systems architecture 

6. An illustrative example 

In this section we will demonstrate how to produce an agent for the ‘Intelligent Agents in 
VRML Worlds’ framework in a step by step manner. In fact, the result will be the same agent 
as the one initially used in the first demonstrative implementation of the framework. 

The first step is the identification of the desired agent’s abilities in a conceptual layer, a 
‘requirement analysis’ task. The agent’s task is to find two keys, unlock two locks and 
eventually exit the maze. In order to perform these tasks, the following actions are needed. At 
first, the agent needs to move from a location to it’s neighboring ones, if a passageway is 
available. Then, the agent needs to pickup things but can carry only one item at a time. 
Furthermore, the agent can unlock locks if he possesses the appropriate key. Finally, the 
agent can leave the maze through its exit, if all locks are unlocked. 

The next development task is to construct VAL code defining the above abilities. A VAL 
program that does this is shown in figure 6.2. 

After VAL code is constructed, an Agent implementation can be produced using the VAL 
compiler. For the code in figure 6.2 the resulting Prolog code is shown in figure 6.3. For 
reasons of comparison, the initial, manually created Prolog code is shown in figure 6.4. 

Finally, the agent can be executed. The results obtained from the execution of the VAL based 
agent are exactly the same with the ones of the initial agent, meaning that the desired 
behavior is achieved. Snapshots from execution of the agent defined in 6.2 are shown in 
figure 6.1 below. 

 

- Object Visualisation 

- Action Visualisation 

- Abstract Ability Definitions 

- Logical Processes Definitions 

- Detailed Ability Definitions 

- Domain Specific Instantiations 

Upper Module 

Intermediate Module 

Lower Module 

VAL 

Agent 
Components 

VRML code, 
object 



 
Initial world state 

 
 

Applying a ‘pickup’ action 

 
 

The agent is at the exit 

Figure 6.1 – Maze agent execution snapshots 

In the first snapshot the world is at its initial state. The agent, located on the upper left room, 
can now start its planning which is based on the abilities it knows it has. These are the 
abilities defined in 6.2. In the second snapshot we the agent encounters a blue key. Since the 
agent is capable of picking up items and the specific item is necessary for completing a goal, 
the agent will apply the ‘pickup’ action on the key. Finally, in the third snapshot, the agent 
reaches the previously encountered exit. The agent reached this point after some planning 
based on the knowledge it has about its actions. The agent knows that it can ‘go’ to some 
locations and is also aware of the exit’s location. The result of the planning process will be a 
series of ‘go’ actions which will lead the agent to the exit. 

It is obvious that apart of the actual ability definitions VAL code affects the agent’s 
decisions, since it provides knowledge about what the agent is able to do. Planning is based 
on reasoning about these abilities, while actions are the actual applications of abilities on 
certain circumstances. 

 

agent adventurer { 
 abilities { 
  go(NewPos) { 
   preconditions { 
    at(Pos), 
    connects(Pos, NewPos) 
   } 
   effects { 
    \- at(Pos), 
    \+ at(NewPos) 
   } 
  } 
  pickup(Item) { 
   preconditions { 
    holding([]), 
    at(Pos), 
    is_at(Item , Pos) 
   } 
   effects { 
    at(Pos), 



    - is_at(Item, Pos), 
    \- holding([]), 
    \+ holding([Item]) 
   } 
  } 
  unlock(Lock) { 
   preconditions { 
    holding([Key]), 
    key(Key), 
    lock(Lock), 
    locks_key_is(Lock, Key), 
    at(Pos), 
    is_at(Lock, Pos) 
   } 
   effects { 
    at(Pos), 
    holding([Key]), 
    - is_at(Lock, Pos), 
    + is_at(Lock, 0), 
    \- holding([Key]), 
    \+ holding([]) 
   } 
  } 
 } 
} 

Figure 6.2 – VAL code for the agent in the maze 

7. Conclusions, implementation details and future work 

In this paper we have presented VAL, an implementation-independent agent-oriented 
programming language, VAL-based systems architecture and a VAL compiler for the 
‘Intelligent Agents in VRML Worlds’ framework. 

The compiler was implemented as a Win32 application written in C++. Parsing-related code 
was produced by ‘Parser Generator’, a Windows based Yacc and Lex IDE by P. D. Stearns. 
For the low-level storage of agent specific information several specialized classes and 
libraries, such as the CAgent class were created. These can be used towards extending VAL 
as well as for production of VAL compilers for other agent systems. 

Future work in the field includes extending VAL to make it a complete agent definition 
language. Apart from an agent’s abilities, several other aspects need to be defined, such as 
it’s goals and reasoning abilities. Moreover, we intend to enrich VAL with temporal logic 
representation capabilities [12, 13, 14]. Due to the open and modular character of VAL, these 
are quite reachable targets. 

Furthermore, future plans include the creation of VAL compilers for other agent systems, so 
that the purpose of high-level agent definition code portability is served. Also, the 
development of an agent testbed environment based on a more elegant and open architecture 
is intended. The target is to provide a complete agent development and execution 
environment that will serve the purposes of research and application on real world domains. 

